
Abstract. We show that the set-theoretic semantics for λ→2 is complete
by inverting evaluation using decision trees. This leads to an implemen-
tation of normalization by evaluation which is witnessed by the source of
part of this paper being a literate Haskell script. We show the correctness
of our implementation using logical relations.

1 Introduction

Which is the simplest typed λ-calculus without uninterpreted base types or type
variables? We suggest that the answer should be λ→2: simply typed lambda
calculus extended by the type of booleans Bool with True, False : Bool and
If t u0 u1 : σ, given t : Bool and u0, u1 : σ. The equational theory is given by
the usual βη-equations of λ→ and the following equations concerning Bool:

If Trueu0 u1 =βη u0

If Falseu0 u1 =βη u1

If t True False =βη t

v (If t u0 u1) =βη If t (v u0) (v u1)

The equations are motivated by the categorical interpretation of Bool as a
boolean object, i.e., an object Bool such that Hom(Γ ×Bool, A) ' Hom(Γ,A)×
Hom(Γ,A) (naturally in Γ and A). The calculus can thus be interpreted in any
cartesian closed category with Bool (using the cartesian structure to interpret
contexts).

The equational theory introduces some interesting equalities. E.g., consider

once = λf : Bool→ Bool. λx : Bool. f x

thrice = λf : Bool→ Bool. λx : Bool.f (f (f x))

We observe that once =βη thrice. To see this, we note that, given f : Bool →
Bool, we have

f (f (f True)) =βη If (f True) (f (f True)) (f (f False))
=βη If (f True) True (f (f False))
=βη If (f True) True (If (f False) (f True) (f False))
=βη If (f True) True (If (f False) False False)
=βη If (f True) True False
=βη f True

Symmetrically, we can show that f (f (f False)) =βη f False, and hence

thrice

= λf : Bool→ Bool. λx : Bool. f (f (f x))
=βη λf : Bool→ Bool. λx : Bool. Ifx (f (f (f True))) (f (f (f False)))
=βη λf : Bool→ Bool.λx : Bool. Ifx (f True) (f False)
=βη λf : Bool→ Bool. λx : Bool.f x

= once

It is easy to see that once and thrice are equal in the standard semantics where
Bool is interpreted by a two-element set Bool = {true, false} and function types
are set-theoretic function spaces. We observe that there are only four elements
in Bool → Bool = {x 7→ true, x 7→ x, x 7→ ¬x, x 7→ false} and that for all the
four f ∈ Bool→ Bool we have f3 = f .

May we use set-theoretic reasoning to prove equalities up to βη-convertibility?
The answer is yes for λ→2, because for λ→2 we can invert set-theoretic evaluation
of typed closed terms. That is: we can define a function quoteσ ∈ JσKset → Tm σ
such that t =βη quoteσ JtKset, for any t ∈ Tm σ. Consequently, we get that, for
any t, t′ ∈ Tm σ, t =βη t′ ⇐⇒ JtKset = Jt′Kset.

The existence of quote also implies that =βη is maximally consistent, i.e.,
identifying any two non-βη-convertible closed terms would lead to an inconsistent
theory. This provides another justification for the specific choice of =βη.

We do not analyze the normal forms here, i.e. the codomain of nf and quote
here. However, the construction presented here, which is based on decision trees,
leads to simple normal forms and we conjecture that this is the same set as the
set of normal forms presented in [1, 5] in the case Bool = 1 + 1.

Haskell as a poor man’s Type Theory

Our construction is entirely constructive, so it can be carried out, e.g., in Martin-
Löf’s Type Theory, and we obtain an implementation of normalization nfσ t =
quoteσ JtKset. We shall here use the functional language Haskell as a poor man’s
Type Theory and obtain a Haskell program to normalize terms.

Haskell hasn’t got dependent types (in particular inductive families), hence
the Haskell types we are using are only approximations of their type-theoretic



correspondents. E.g., the type-theory type TmΓ σ contains all welltyped terms
of type σ in context Γ , but its Haskell counterpart Tm contains all untyped terms.
Similarly, the set-theoretic denotation of a type σ :→ τ is given by Jσ :→ τKset =
JσKset → JτKset but its Haskell implementation is given by a recursive type El
with a constructor SLam ∈ Ty→ (El→ El)→ El.

We believe that this informal use of Type Theory is an effective way to
arrive at functional programs which are correct by construction. However, we
hope that in the future we can go further and bridge the gap between informal
type theoretic reasoning and the actual implementation by using a dependently
typed programming language. such as the Epigram system, which is currently
being developed by Conor McBride [12].

Related work

Inverting evaluation to achieve normalization by evaluation (NBE, aka. reduction-
free normalization) was pioneered in [6] for simply typed lambda calculus with
type variables and a non-standard semantics; a categorical account in terms of
presheaves was given in [2]; this was extended to System F in [3, 4]; see [9] for
a recent survey on NBE. The completeness of the set-theoretic model in the
presence of coproducts has been shown in [8] and our case arises as a special
case when there are no type variables. Normalization procedures for typed λ-
calculus with coproducts can be found in [10, 11] using rewriting techniques and
[1, 5] using NBE and sheaf theory. Both approaches allow type variables but do
not handle the empty type. Here we present a much simpler construction for
closed types using the simplest possible semantics of first-order simply typed
λ-calculi—the set-theoretic one—and also provide a concrete implementation of
quote and nf, whose correctness we show in detail.

2 Implementation of λ→2

The source of Sections 2 and 3 of this paper is a literate Haskell script imple-
menting normalization for λ→2 and is available from

http://www.cs.nott.ac.uk/~txa/publ/Nbe2.lhs

We start by introducing types Ty ∈ ?, variables Var ∈ ?, typing contexts
Con ∈ ? and untyped terms Tm ∈ ? of the object language by the following
Haskell datatype definitions:

data Ty = Bool | Ty :-> Ty
deriving (Show, Eq)

type Var = String

type Con = [ (Var, Ty) ]



data Tm = Var Var
| TTrue | TFalse | If Tm Tm Tm
| Lam Ty String Tm | App Tm Tm
deriving (Show, Eq)

We view these recursive definitions as inductive definitions, i.e., we do not con-
sider infinite terms. All the functions we define are total wrt. their precise type-
theoretic types.

Implementing typed terms Tm ∈ Con → Ty → ? would take inductive fam-
ilies, which we cannot use in Haskell. But we can implement type inference
infer ∈ Con→ Tm→ Maybe Ty (where Maybe X = 1 + X as usual):

infer :: Con -> Tm -> Maybe Ty
infer gamma (Var x) =

do sigma <- lookup x gamma
Just sigma

infer gamma TTrue = Just Bool
infer gamma TFalse = Just Bool
infer gamma (If t u0 u1) =

do Bool <- infer gamma t
sigma0 <- infer gamma u0
sigma1 <- infer gamma u1
if sigma0 == sigma1 then Just sigma0 else Nothing

infer gamma (Lam sigma x t) =
do tau <- infer ((x, sigma) : gamma) t

Just (sigma :-> tau)
infer gamma (App t u) =

do (sigma :-> tau) <- infer gamma t
sigma’ <- infer gamma u
if sigma == sigma’ then Just tau else Nothing

This implementation is correct in the sense that t ∈ TmΓ σ iff inferΓ t = Just σ.
Evaluation of types J−K ∈ Ty → ? is again an inductive family, which we

cannot implement in Haskell, and the workaround is to have all JσK coalesced
into one metalanguage type el (of untyped elements) much the same way as all
TmΓ σ appear coalesced in Tm. We use a type class Sem to state what we require
of such a coalesced type el :

class Sem el where
true :: el
false :: el
xif :: el -> el -> el -> el
lam :: Ty -> (el -> el) -> el
app :: el -> el -> el

Evaluation of types J−K ∈ Ty → ? naturally induces evaluation of contexts
J−K ∈ Con→ ?, defined by



[] ∈ J[]K
ρ ∈ JΓ K d ∈ JσK

(x, d) : ρ ∈ J(x, σ) : Γ K

We write Tm Γ = JΓ Ksyn for the set of closed substitutions, which arises as
in instance when using JσKsyn = Tm[]σ as the interpretation of types.

In the Haskell code we approximate evaluation of contexts by Env:

type Env el = [ (Var, el) ]

Given t ∈ TmΓ σ we define the evaluation of terms JtK ∈ JΓ K → JσK. In Haskell
this is implemented as eval:

eval :: Sem el => Env el -> Tm -> el
eval rho (Var x) = d

where (Just d) = lookup x rho
eval rho TTrue = true
eval rho TFalse = false
eval rho (If t u0 u1) =

xif (eval rho t) (eval rho u0) (eval rho u1)
eval rho (Lam sigma x t) =

lam sigma (\ d -> eval ((x, d) : rho) t)
eval rho (App t u) = app (eval rho t) (eval rho u)

The standard set-theoretic semantics is given by

JBoolKset = Bool

Jσ :→ τKset = JσKset → JτKset

This can be represented in Haskell as an instance of Sem:

data El = STrue | SFalse | SLam Ty (El -> El)

instance Sem El where
true = STrue
false = SFalse
xif STrue d _ = d
xif SFalse _ d = d
lam = SLam
app (SLam _ f) d = f d

Since sets form a cartesian closed category with a boolean object, the set-
theoretic semantics validates all βη-equalities. This is to say that J−Kset is equa-
tionally sound:

Proposition 2.1 (Soundness). If ρ ∈ JΓ K and t =βη t′ ∈ TmΓ σ, then
JtKset ρ = Jt′Kset ρ.

Since all the sets we consider are finite, semantic equality can be implemented
in Haskell, by making use of the function enum ∈ (σ ∈ Ty) → Tree JσK, which
we will provide later:



instance Eq El where
STrue == STrue = True
SFalse == SFalse = True
(SLam sigma f) == (SLam _ f’) =

and [f d == f’ d | d <- flatten (enum sigma)]
_ == _ = False

Using on the same function we can also print elements of El:

instance Show El where
show STrue = "STrue"
show SFalse = "SFalse"
show (SLam sigma f) =

"SLam " ++ (show sigma) ++ " " ++
(show [ (d, f d) | d <- flatten (enum sigma) ])

The equational theory of the calculus itself gives rise to another semantics—
the free semantics, or typed terms up to βη-convertibility. This can be approxi-
mated by the following Haskell code, which uses a redundancy-avoiding version
if′ of If which produces a shorter but βη-equivalent term:

if’ :: Tm -> Tm -> Tm -> Tm
if’ t TTrue TFalse = t
if’ t u0 u1 = if u0 == u1 then u0 else If t u0 u1

instance Sem Tm where
true = TTrue
false = TFalse
xif = if’
lam sigma f = Lam sigma "x" (f (Var "x"))
app = App

We also observe that the use of a fixed variable is only justified by the fact that
our algorithm uses at most one bound variable at the time. A correct dependently
typed version of the free semantics requires the use of presheaves to ensure that
the argument to Lam is stable under renaming. We refrain from presenting the
details here. It is well known that this semantics is equationally sound.

3 Implementation of quote

We now proceed to implementing quote ∈ (σ ∈ Ty)→ JσKset → Tm σ.
To define quoteσ→τ we use enumσ, which generates a decision tree whose

leaves are all the elements of JσK, and questionsσ, which generates the list of
questions, i.e. elements of JσK→ JBoolK, based on answers to whom an element
of JσK can be looked up in the tree enumσ. (Since our decision trees are perfectly
balanced and we use the same list questions along each branch of a tree, we
separate the questions labelling from the tree.)

Decision trees Tree ∈ Ty→ ? are provided by



data Tree a = Val a | Choice (Tree a) (Tree a) deriving (Show, Eq)

We will exploit the fact that Tree is a monad

instance Monad Tree where
return = Val
(Val a) >>= h = h a
(Choice l r) >>= h = Choice (l >>= h) (r >>= h)

(return and >>= are Haskell for the unit resp. the bind or Kleisli extension
operation of a monad) and hence a functor

instance Functor Tree where
fmap h ds = ds >>= return . h

(fmap is Haskell for the action of a functor on morphisms).
It is convenient to use the function flatten which calculates the list of leaves

of a given tree:

flatten :: Tree a -> [ a ]
flatten (Val a) = [ a ]
flatten (Choice l r) = (flatten l) ++ (flatten r)

We implement enumσ and questionsσ by mutual induction on σ ∈ Ty. The
precise typings of the functions are enum ∈ (σ ∈ Ty)→ Tree JσK and questions ∈
(σ ∈ Ty) → [JσK → JBoolK]. As usual, Haskell cannot express those subtleties
due to its lack of dependent types, but we can declare

enum :: Sem el => Ty -> Tree el
questions :: Sem el => Ty -> [ el -> el ]

The base case is straightforward: A boolean is true or false and to know
which one it is it suffices to know it.

enum Bool = Choice (Val true) (Val false)

questions Bool = [ \ b -> b ]

The implementation of enumσ:→τ and questionsσ:→τ proceeds from the idea
that a function is determined by its graph: to know a function it suffices to know
its value on all possible argument values. The main idea in the implementation
of enumσ:→τ is therefore to start with enumτ and to duplicate the tree for each
question in questionsσ using the bind of Tree:

enum (sigma :-> tau) =
fmap (lam sigma) (mkEnum (questions sigma) (enum tau))

mkEnum :: Sem el => [ el -> el ] -> Tree el -> Tree (el -> el)
mkEnum [] es = fmap (\ e -> \ d -> e) es
mkEnum (q : qs) es = (mkEnum qs es) >>= \ f1 ->

(mkEnum qs es) >>= \ f2 ->
return (\ d -> xif (q d) (f1 d) (f2 d))



questionsσ:→τ produces the appropriate questions by enumerating σ and using
questions from τ :

questions (sigma :-> tau) =
[ \ f -> q (app f d) | d <- flatten (enum sigma),

q <- questions tau ]

As an example, the enumeration and questions for Bool :→ Bool return:

Choice
(Choice

(Val (lam Bool (\ d -> xif d true true)))
(Val (lam Bool (\ d -> xif d true false))))

(Choice
(Val (lam Bool (\ d -> xif d false true )))
(Val (lam Bool (\ d -> xif d false false))))

resp.

(\ f -> app f true :
(\ f -> app f false :

[]))

We can look up an element in the decision tree for a type by answering all
the questions, this is realized by the function find below. To define the domain of
find precisely we define a relation between lists of questions and trees of answers
� ⊆ [a]× Tree b inductively:

[] � (Val t)
as � l as � r

a : as � Choice l r

Now given as ∈ [JBoolK], ts ∈ Tree JσK, s.t. as � ts we obtain find as ts ∈ JσK,
implemented in Haskell:

find :: Sem el => [ el ] -> Tree el -> el
find [] (Val t) = t
find (a : as) (Choice l r) = xif a (find as l) (find as r)

We are now ready to implement quoteσ ∈ JσKset → Tm σ, with Haskell typing

quote :: Ty -> El -> Tm

by induction on σ ∈ Ty. As usual, the base case is easy:

quote Bool STrue = TTrue
quote Bool SFalse = TFalse

quoteσ:→τ is more interesting: Our strategy is to map quoteτ ◦ f to the set-
theoretic enumτ and to then build a tree of If expressions by using the syntactic
questionsσ in conjunction with the syntactic find:



quote (sigma :-> tau) (SLam _ f) =
lam sigma (\ t -> find [ q t | q <- questions sigma ]

(fmap (quote tau . f) (enum sigma)))

(Notice that in Haskell it is inferred automatically which semantics is meant
where.)

As already discussed in the introduction, we implement normalization nf ∈
(σ ∈ Ty)→ Tm σ → Tm σ by

nf :: Ty -> Tm -> Tm
nf sigma t = quote sigma (eval [] t)

Since we can infer types, we can implement nf′ ∈ Tm→ Maybe (Σσ∈TyTm σ):

nf’ :: Tm -> Maybe (Ty, Tm)
nf’ t = do sigma <- infer [] t

Just (sigma, nf sigma t)

We test our implementation with the example from the introduction:

b2b = Bool :-> Bool
once = Lam b2b "f" (Lam Bool "x" (App (Var "f") (Var "x")))
twice = Lam b2b "f" (Lam Bool "x" (App (Var "f")

(App (Var "f") (Var "x"))))
thrice = Lam b2b "f"

(Lam Bool "x" (App (Var "f")
(App (Var "f")

(App (Var "f") (Var "x")))))

and convince ourselves that (nf′ once == nf′ thrice) = true but (nf′ once ==
nf′ twice) = false. Since semantic equality is decidable we do not actually have
to construct the normal forms to decide convertibility.

Since testing can only reveal the presence of errors we shall use the rest of
this paper to prove that quote and hence nf behave correctly.

4 Correctness of quote

The main tool in our proof will be a notion of logical relations, a standard tool
for the characterization of definable elements in models of typed lambda calculi
since the pioneering work of Plotkin [13].

Let us agree to abbreviate Tm[] σ by Tm σ and JtKset[] by JtKset.

Definition 4.1 (Logical Relations). We define a family of relations Rσ ⊆
Tm σ × JσKset by induction on σ ∈ Ty as follows:

– tRBoolb iff t =βη True and b = true or t =βη False and b = false;
– tRσ:→τf iff, for all u, d, uRσd implies App t uRτf d.



Note that R is not indexed by contexts, logical relations only relate closed terms.
We extend logical relations to contexts: Given Γ ∈ Con we define RΓ ⊆

TmΓ × JΓ K by:

[]R[][]
ρRΓ ρ′ dRσd′

(x, d) : ρ R(x,σ):Γ (x, d′) : ρ′

Logical relations are invariant under βη-equality.

Lemma 4.2. If tRσd and t =βη t′, then t′Rσd.

Logical relations obey the following Fundamental Theorem, a kind of sound-
ness theorem for logical relations.

Lemma 4.3 (Fundamental Theorem of Logical Relations). If θRΓ ρ and
t ∈ TmΓ σ, then [t] θRσJtKset ρ. In particular, if t ∈ Tm σ, then tRσJtKset.

The main result required to see that quote is correct is the following lemma:

Lemma 4.4 (Main Lemma). If tRσd, then t =βη quoteσ d.

The proof of this lemma is the subject of the next section.
By correctness of quote we mean that it inverts set-theoretic evaluation of

typed closed terms.

Theorem 4.5 (Main Theorem). If t ∈ Tm σ, then t =βη quoteσ JtKset.

Proof. Immediate from the Fundamental Theorem and the Main Lemma. ut

The (constructive) existence and correctness of quote has a number of straight-
forward important consequences.

Corollary 4.6 (Completeness). If t, t′ ∈ Tm σ, then JtKset = Jt′Kset implies
t =βη t′.

Proof. Immediate from the Main Theorem. ut

From soundness (Proposition 2.1) and completeness together we get that =βη is
decidable: checking whether t =βη t′ reduces to checking whether JtKset = Jt′Kset,
which is decidable as J−Kset is computable and equality in finite sets is decidable.

Corollary 4.7. If t, t′ ∈ Tm σ, then t =βη t′ iff quoteσ JtKset = quoteσ Jt′Kset.

Proof. Immediate from soundness (Proposition 2.1) and the Main Theorem. ut

This corollary shows that nfσ = quoteσ ◦ J−Kset : Tm σ → Tm σ indeed makes
sense as normalization function: apart from just delivering, for any given typed
closed term, some βη-equal term, it is actually guaranteed to deliver the same
term for t, t′, if t, t′ are βη-equal (morally, this is Church-Rosser for reduction-
free normalization).

Note that although we only stated completeness and normalization for typed
closed terms above, these trivially extend to all typed terms as opens terms can
always be closed up by lambda-abstractions and this preserves βη-equality.



Corollary 4.8. If t, t′ ∈ Tm σ and [C] [(x, t)] =βη [C] [(x, t′)] for every C :
Tm[(x,σ)] Bool, then t =βη t′. Or, contrapositively, and more concretely, if t, t′ ∈
Tm (σ1 :→ . . . :→ σn :→ Bool) and t 6=βη t′, then there exist u1 ∈ Tm σ1,
. . .un ∈ Tm σn such that

nfBool (App (. . . (App t u1) . . .) un) 6= nfBool (App (. . . (App t′ u1) . . .) un)

Proof. This corollary does not follow from the statement of the Main Theorem,
but it follows from its proof. ut

Corollary 4.9 (Maximal consistency). If t, t′ ∈ Tm σ and t 6=βη t′, then
from the equation t = t′ as an additional axiom one would derive True = False.

Proof. Immediate from the previous corollary. ut

5 Proof of the main lemma

We now present the proof of the main lemma which was postponed in the previ-
ous section. To keep the proof readable, we write enumset, questionsset, findset to
emphasize the uses of the set-theoretic semantics instances of enum, questions,
find, while the free semantics instances will be written as enumsyn, questionssyn,
findsyn. We use the fact that any functor F such as Tree has an effect on relations
R ⊆ A×B denoted by FR ⊆ FA× FB, which can be defined as:

z ∈ F{(a, b) ∈ A×B | aRb}

fmap fst z FR fmap snd z

We first give the core of the proof and prove the lemmas this takes afterwards.

Proof (of the Main Lemma). By induction on σ.

– Case Bool: Assume tRBoolb. Then either t =βη True and b = true, in which
case we have

t =βη True = quoteBool true = quoteBool b

or t =βη False and b = false, in which case we have

t =βη False = quoteBool false = quoteBool b

– Case σ :→ τ : Assume tRσ:→τf , for all u, d, uRσd implies App t uRτf d. We
have

t =βη Lamσ x (App t (Var x))
=βη (by Lemma 5.2 below)

Lamσ x (App t (findsyn [q (Var x) | q ← questionsσ
syn] enumσ

syn))
= Lamσ x (findsyn [q (Var x) | q ← questionsσ

syn]
(fmap (App t) enumσ

syn))
=βη (by Sublemma)

Lamσ x (findsyn [q (Var x) | q ← questionsσ
syn]

(fmap (quoteτ ◦ f) enumσ
set))

= quoteσ:→τ f



The Sublemma is:

fmap (App t) enumσ
syn (Tree =βη) fmap (quoteτ ◦ f) enumσ

set

For proof, we notice that, by Lemma 5.1 (1) below for σ,

enumσ
syn (Tree Rσ) enumσ

set

Hence, by assumption and the fact that fmap commutes with the effect on
relations

fmap (App t) enumσ
syn (Tree Rτ ) fmap f enumσ

set

Hence, by IH of the Lemma for τ ,

fmap (App t) enumσ
syn (Tree =βη) fmap (quoteτ ◦ f) enumσ

set

ut

The proof above used two lemmas. One is essentially free, but the other is
technical.

Lemma 5.1 (“Free” Lemma).

1. enumσ
syn (Tree Rσ) enumσ

set.
2. questionsσ

syn [Rσ → RBool] questionsσ
set.

Proof. The proof is simultaneous for (1) and (2) by induction on σ.

– Case Bool: Trivial.
– Case σ :→ τ : Proof of (1) uses IH (2) for σ and IH (1) for τ ; proof of (2)

uses IH (1) for σ and IH (2) for τ .
ut

Lemma 5.2 (Technical Lemma). For t ∈ TmΓ σ:

t =βη findsyn [q t | q ← questionsσ
syn] enumσ

syn

Proof. By induction on σ.

– Case Bool:

t = If′ t True False

= If′ t (findsyn [] (Val True)) (findsyn [] (Val False))
= findsyn [t] (Choice (Val True) (Val False))
= findsyn [q t | q ← questionsBoolsyn ] enumBool

syn



– Case σ :→ τ :

t =βη Lamσ z (App t (Var z)) (z fresh wrt Γ )
=βη (by IH for σ)

Lamσ z (findsyn [q(Var z) | q ← questionsσ
syn] enumσ

syn)
=βη (by Sublemma below)

Lamσ z

(findsyn [q (App t u) | u← flatten enumσ
syn, q ← questionsτ

syn]
(fmap (λg g (Var z)) (mkenumsyn questionsσ

syn enumτ
syn)))

=βη by Lemma 5.3 and functor laws
findsyn [q (App t u) | u← flatten enumσ

syn, q ← questionsτ
syn]

(fmap (lam σ) (mkenumsyn questionsσ
syn enumτ

syn)))
= findsyn [q t | q ← questionsσ:→τ

syn ] enumσ:→τ
syn

The sublemma is: Given qs � us then

App t (findsyn [q u | q ← qs] us)
=βη findsyn [q (App t u′) | u′ ← flatten us, q ← questionsτ

syn]
(fmap (λg g u) (mkenumsyn qs enumτ

syn))

The proof is by induction on qs � us.
• Case [] � Val u?:

Assume u =βη findsyn [] (Val u?), i.e., u =βη u?. We get
App t (findsyn [q u | q ← qs] us)
=βη (by IH of the Lemma for τ)

findsyn [q (App t u?) | q ← questionsτ
syn] enumτ

syn

= findsyn [q (App t u?) | q ← questionsτ
syn]

(fmap (λg g u) (fmap (λvλu′ v) enumτ
syn))

= findsyn [q (App t u′) | u′ ← [u?], q ← questionsτ
syn]

(fmap (λg g u) (mkenumsyn [] enumτ
syn))

• Case q : qs � Choice l r:

App t (findsyn [q′ u | q′ ← q : qs] (Choice l r))
= App t (if ′ (q u) (findsyn [q′ u | q′ ← qs] l)

(findsyn [q′ u | q′ ← qs] r)
=βη if ′ (q u) (App t (findsyn [q′ u | q′ ← qs] l))

(App t (findsyn [q′ u | q′ ← qs] r))
=βη (by IH of the Sublemma for qs � l, qs � r)

if ′ (q u)
(findsyn [q′ (App t u′) | u′ ← flatten l, q′ ← questionsτ

syn]
(fmap (λg g u) (mkenumsyn qs enumτ

syn)))



(findsyn [q′ (App t u′) | u′ ← flatten r, q′ ← questionsτ
syn]

(fmap (λg g u) (mkenumsyn qs enumτ
syn)))

= findsyn [] (Val (If′ (q u)
(findsyn [q′ (App t u′) | u′ ← flatten l, q′ ← questionsτ

syn]
(fmap (λg g u) (mkenumsyn qs enumτ

syn)))
(findsyn [q′ (App t u′) | u′ ← flatten r, q′ ← questionsτ

syn]
(fmap (λg g u) (mkenumsyn qs enumτ

syn)))))
=βη (by twice Lemma 5.4)

findsyn ([q′ (App t u′) | u′ ← flatten l, q′ ← questionsτ
syn]

++ ([q′ (App t u′) | u′ ← flatten r, q′ ← questionsτ
syn]

++ []))
((fmap (λg g u) (mkenumsyn qs enumτ

syn))�= λv0

(fmap (λg g u) (mkenumsyn qs enumτ
syn))�= λv1

Val (If′ (q u) v0 v1))
= findsyn ([q′ (App t u′) | u′ ← flatten l, q′ ← questionsτ

syn]
++ [q′ (App t u′) | u′ ← flatten r, q′ ← questionsτ

syn])
(fmap (λg g u)

((mkenumsyn qs enumτ
syn)�= λg0

(mkenumsyn qs enumτ
syn �= λg1

Val (λu′ If′ (q u′) (g0 u′) (g1 u′))))
= findsyn [q′ (App t u′) | t← (flatten l) ++(flatten r),

q′ ← questionsτ
syn]

(fmap (λg g u) (mkenumsyn (q : qs) enumτ
syn))

= findsyn [q′ (App t u′) | u′ ← flatten (Choice l r),
q′ ← questionsτ

syn]
(fmap (λg g u) (mkenumsyn (q : qs) enumτ

syn))
ut

We have used two lemmas, which are easy to prove:

Lemma 5.3. Given as � us it is true that

findsyn as (fmap f us) =βη f (findsyn as us)

Proof. Simple induction on as � us.

Lemma 5.4. Given as � ts and, for all u ∈ TmΓ σ, bs � h u, then

findsyn (as ++bs) (ts �= h) =βη findsyn bs (h (findsyn as ts))

Proof. By induction on as � ts.



– Case [] � Val t:

findsyn ([] ++bs) ((Val t)�= h)
= findsyn bs ((Val t)�= h)
= findsyn bs (h t)
= findsyn bs (h (findsyn [] (Val t))

– Case a : as � Choice l r:

findsyn ((a : as) ++bs) ((Choice l r)�= h)
= findsyn (a : (as ++bs)) ((Choice l r)�= h)
= If′ a (findsyn (as ++bs) (l�= h)) (findsyn (as ++bs) (r �= h))
= (by IH for as � l, as � r)

If′ a (findsyn bs (h (findsyn as l))) (findsyn bs (h (findsyn as r)))
=βη findsyn bs (h (If′ a (findsyn as l) (findsyn as r)))
= findsyn bs (h (findsyn (a : as) (Choice l r)))

ut

6 Discussion and further work

Instead of decision trees we could have used a direct encoding of the graph of a
function, we call this the truth-table semantics. However, this approach leads not
only too much longer normal forms but also the semantic equality is less efficient.
On the other hand it is possible to go further and use Binary Decision Diagrams
(BDDs) [7] instead of decision trees. We plan to explore this in further work and
also give a detailed analysis of the normal forms returned by our algorithm.

We have argued that λ→2 is the simplest λ-calculus with closed types, how-
ever we are confident that the technique described here works also for closed
types in λ0+1×→ (the finitary λ-calculus). We leave this extension for a journal
version of this work.

One can go even further and implement finitary Type Theory, i.e. λ012ΣΠ

(note that A + B = Σx ∈ 2.if x thenA else B). This could provide an interesting
base for a type-theoretic hardware description and verification language.

The approach presented here works only for calculi without type variables. It
remains open to see whether this approach can be merged with the the standard
techniques for NBE for systems with type variables, leading to an alternative
proof of completeness and maybe even finite completeness for the calculi dis-
cussed above.

